

# Tas, files et plus courts chemins

MPSI/MP OPTION INFORMATIQUE - TP n° 3.2 - Olivier Reynet

## À la fin de ce chapitre, je sais :

- ☞ définir un tas-min et un tas-max
- ☞ trier un tableau pas tas
- ☞ utiliser un tas pour créer une file de priorité
- ☞ appliquer les tas à l'algorithme de Dijkstra

## A Implémentation d'un tas max et tri par tas

- A1. Créer une variable `heap_test` de type `Array` contenant les entiers de 1 à 10 dans l'ordre décroissant. On générera automatiquement cette variable (pas in extenso). Est-ce un tas? Si oui, de quel type?

### Solution :

```
1 let heap_test = Array.init 10 (fun i -> 10 - i);;  
2 val heap_test : int array = [|10; 9; 8; 7; 6; 5; 4; 3; 2; 1|]
```

C'est un tas-max.

- A2. Écrire une fonction de signature `swap : 'a array -> int -> int -> unit` qui échange la place de deux éléments dans un type `Array`. Tester cette fonction sur le tableau `heap_test`.

### Solution :

```
1 let swap t i j = let tmp = t.(i) in t.(i) <- t.(j); t.(j) <- tmp;;
```

Comme en OCaml les `Array` sont indexés à partir de 0, on choisit de placer la racine du tas en 0. Puis, si le père est  $k$  alors les fils seront en  $2k + 1$  et  $2k + 2$ . Si un fils est en  $k$ , alors le père est en  $\lfloor \frac{k-1}{2} \rfloor$ .

Il faut noter qu'un tas peut ne pas être rempli. Dans ce cas, on prendra soin de relever l'indice de la première case non remplie.

- A3. Écrire une fonction de signature `up : 'a array -> int -> unit` qui, si cela est possible, fait monter un élément d'après son indice dans le tas tout en préservant la structure du tas.

### Solution :

```

1 let up heap k =
2   let rec aux i = match i with
3     | 0 -> ()
4     | _ -> let f = (i - 1)/2 in
5             if heap.(i) > heap.(f) then (swap heap i f ; aux f)
6   in aux k;;

```

- A4. Écrire une fonction de signature `down : 'a array -> int -> int -> unit` qui, si cela est possible, fait descendre un élément dans un tas max tout en préservant la structure du tas. Le premier entier de la signature est l'indice de la première case non remplie du tas, le second l'indice de l'élément à faire descendre.

**Solution :**

```

1 let down heap first_not_used k =
2   let rec aux i = match i with
3     | i when 2*i + 1 > first_not_used -> () (* Leave *)
4     | i when 2*i + 2 = first_not_used -> let f = 2*i + 1 in
5                                           if heap.(i) < heap.(f)
6                                           then (swap heap i f; aux f;)
7     | i -> begin
8       let f = if heap.(2*i + 1) > heap.(2*i + 2)
9               then 2*i + 1 else 2*i + 2 in
10      if heap.(i) < heap.(f)
11      then (swap heap i f; aux f;)
12      end
13   in aux k;;

```

On cherche à créer un tas de deux manière différente :

- soit en considérant que le premier élément du tableau est un tas que l'on fait grossir avec les éléments de droite du tableau,
  - soit en considérant que les feuilles sont des tas à un élément que l'on fait grossir au fur avec les éléments de gauche du tableau. On remarque dans ce cas que quelle que soit la taille  $n$  du tas, l'élément d'indice  $n/2$  est toujours la première feuille. On va donc insérer les  $n/2$  premiers éléments dans leurs feuilles et faire apparaître la structure de tas.
- A5. Effectuer à la main la transformation du tableau `[|0; 1; 2; 3; 4; 5; 6; 7; 8; 9|]` en tas-max par les deux méthodes.
- A6. Écrire une fonction de signature `heap_make_up : 'a array -> unit` qui transforme un tableau en un tas en faisant monter les éléments. Quelle la complexité de cette fonction?

**Solution :**

```

1 let heap_make_up t = for k = 1 to Array.length t - 1 do up t k done;;

```

La complexité au pire de cette fonction est linéaire en  $O(n \log n)$  si  $n$  est la taille du tas. En effet, dans le pire des cas, on fait remonter les  $n$  nœuds jusqu'à la racine, c'est à dire toute la hauteur de l'arbre qui vaut  $\log n$ .

A7. Écrire une fonction de signature `heap_make_down : 'a array -> unit` qui transforme un tableau en un tas en faisant descendre les éléments. Quelle est la complexité de cette fonction<sup>1</sup> ?

**Solution :**

```
1 let heap_make_down t = let size = Array.length t in
2   for k = size/2 - 1 downto 0 do down t size k done;;
```

En considérant que le nombre de nœuds ayant la hauteur  $h$  est majoré par  $\lceil \frac{n}{2^{h+1}} \rceil$  on montre que :

$$\sum_{h=1}^p \lceil \frac{n}{2^{h+1}} \rceil h \leq \frac{p(p+1)}{2} + n \sum_{h=1}^p \frac{h}{2^{h+1}} = O(n) \quad (1)$$

A8. Écrire une fonction de signature `heap_sort : 'a array -> unit` qui effectue le tri par tas d'un tableau. Quelle la complexité de cette fonction ?

**Solution :**

```
1 let heap_sort t =
2   heap_make_down t;
3   let size = Array.length t in
4   for k = size - 1 downto 0 do (swap t 0 k; down t k 0) done;;
```

A9. Tester le tri par tas sur un tableau de 10000 valeurs entières choisies aléatoirement entre 0 et 1000000.

**Solution :**

```
1 Random.init 42;;
2 let test = Array.init 100000 (fun i -> Random.full_int 1000000);;
3 heap_sort test;;
4 test;;
```

## B Implémentation d'une file de priorités

Pour implémenter une file de priorités à partir d'un tas, il est nécessaire de définir le type de données que contient la file, puis de programmer les opérations de base de la file de priorités. Les opérations sur une file de priorités sont :

1. créer une file vide,
2. insérer dans la file une valeur associée à une priorité (ENFILER),
3. sortir de la file la valeur associée la priorité maximale (ou minimale) (DÉFILER).

---

1. On suppose que le nombre de nœuds ayant la hauteur  $h$  est majoré par  $\lceil \frac{n}{2^{h+1}} \rceil$

On considère une file de priorités dont les éléments sont des couples  $(v, p)$  et  $p$  est un entier naturel.  
**Plus  $p$  est faible, plus la priorité est élevée.**

Par ailleurs, on se dote des types et des exceptions suivantes :

```
1 type 'a qdata = {value: 'a; priority: int};;
2 type 'a priority_queue = {mutable first_free: int; heap: 'a qdata array};;
3 exception EMPTY_PRIORITY_QUEUE;;
4 exception FULL_PRIORITY_QUEUE;;
```

B1. De quel type est le tas nécessaire à la construction de cette file de priorités?

**Solution :** Il s'agit d'un tas-min.

B2. Écrire une fonction de signature `up : 'a array -> int -> unit` qui, si cela est possible, fait monter un élément d'après son indice dans le tas tout en préservant la structure du tas. On veillera à bien faire monter en fonction de la priorité.

**Solution :**

```
1 let up heap k =
2   let rec aux i = match i with
3     | 0 -> ()
4     | _ -> let p = (i - 1)/2 in
5             if heap.(i).priority < heap.(p).priority
6             then (swap heap i p ; aux p)
7   in aux k;;
```

B3. Écrire une fonction de signature `down : 'a array -> int -> int -> unit` qui, si cela est possible, fait descendre un élément dans un tas max tout en préservant la structure du tas. Le premier entier de la signature est l'indice de la première case non remplie du tas, le second l'indice de l'élément à faire descendre. On veillera à bien faire descendre en fonction de la priorité.

**Solution :**

```
1 let down heap first_not_used k =
2   let rec aux i = match i with
3     | i when 2*i + 1 > first_not_used -> () (* Leave *)
4     | i when 2*i + 2 = first_not_used -> let f = 2*i + 1 in
5             if heap.(i).priority > heap.(f).priority then (swap heap i f;
6               aux f;)
7     | i -> begin
8       let f = if heap.(2*i + 1).priority < heap.(2*i + 2).priority
9               then 2*i + 1 else 2*i + 2 in
10      if heap.(i).priority > heap.(f).priority
11      then (swap heap i f; aux f;)
12    end
13   in aux k;;
```

- B4. Écrire une fonction de signature `make_priority_queue : int -> 'a * int -> 'a priority_queue` qui crée une file de priorité à  $n$  éléments initialisés à l'aide d'un type `qdata` donné en paramètre et dont le premier élément libre est le premier du tas.

**Solution :**

```

1 let make_priority_queue n (v,p) = {
2     first_free = 0;
3     heap = Array.init n (fun i -> {value = v; priority=p});
4 };;

```

- B5. Écrire une fonction de signature `insert : 'a priority_queue -> 'a * int -> unit` qui enfila un élément dans la file de priorités. On veillera à préserver la structure du tas et à renvoyer une exception si l'opération n'est pas possible.

**Solution :**

```

1 let insert pq (v,p) =
2   let size = Array.length pq.heap in
3   if pq.first_free + 1 > size then raise FULL_PRIORITY_QUEUE;
4   pq.heap.(pq.first_free) <- {value=v; priority=p};
5   up pq.heap pq.first_free;
6   pq.first_free <- pq.first_free + 1;;

```

- B6. Écrire une fonction de signature `get_min : 'a priority_queue -> 'a` qui renvoie l'élément `value` de priorité maximale. On veillera à préserver la structure du tas et à renvoyer une exception si l'opération n'est pas possible.

**Solution :**

```

1 let get_min pq =
2   if pq.first_free = 0 then raise EMPTY_PRIORITY_QUEUE;
3   let first = pq.heap.(0).value in
4   pq.first_free <- pq.first_free - 1;
5   pq.heap.(0) <- pq.heap.(pq.first_free);
6   down pq.heap pq.first_free 0;
7   first;;

```

## C Dijkstra et files de priorités

L'algorithme de Dijkstra (cf. figure 1) se décompose en deux opérations répétées à chaque itération :

1. transférer de l'élément de distance minimale dans l'ensemble des éléments explorés,
2. mettre à jour les distances en fonction de ce nouvel élément sur le chemin.

L'utilisation d'une file de priorité afin d'extraire l'élément de distance minimale permet d'améliorer la complexité de l'algorithme et d'atteindre  $O((n + m) \log n)$ , si  $n$  est l'ordre du graphe et  $m$  le nombre

d'arêtes. On cherche donc à implémenter l'algorithme de Dijkstra sur un graphe pondéré en utilisant une file de priorité dont les priorités sont les distances au sommet de départ de l'algorithme.

---

**Algorithme 1** Algorithme de Dijkstra, plus courts chemins à partir d'un sommet donné
 

---

```

1: Fonction DIJKSTRA( $G = (V, E, w), a$ )           ▷ Trouver les plus courts chemins à partir de  $a \in V$ 
2:    $\Delta \leftarrow a$                                ▷  $\Delta$  est l'ensemble des sommets dont on connaît la distance à  $a$ 
3:    $\Pi \leftarrow$  un dictionnaire vide                 ▷  $\Pi[s]$  est le parent de  $s$  dans le plus court chemin de  $a$  à  $s$ 
4:    $d \leftarrow$  l'ensemble des distances au sommet  $a$ 
5:    $\forall s \in V, d[s] \leftarrow w(a, s)$                 ▷  $w(a, s) = +\infty$  si  $s$  n'est pas voisin de  $a$ , 0 si  $s = a$ 
6:   tant que  $\bar{\Delta}$  n'est pas vide répéter             ▷  $\bar{\Delta}$  : sommets dont la distance n'est pas connue
7:     Choisir  $u$  dans  $\bar{\Delta}$  tel que  $d[u] = \min(d[v], v \in \bar{\Delta})$  ▷ On prend la plus courte distance à  $a$  dans  $\bar{\Delta}$ 
8:      $\Delta = \Delta \cup \{u\}$                            ▷ Transfert
9:     pour  $x \in \bar{\Delta}$  répéter                       ▷ Ou bien  $x \in \mathcal{N}_G(u) \cap \bar{\Delta}$ , pour tous les voisins de  $u$  dans  $\bar{\Delta}$ 
10:      si  $d[x] > d[u] + w(u, x)$  alors
11:         $d[x] \leftarrow d[u] + w(u, x)$                ▷ Mises à jour des distances des voisins
12:         $\Pi[x] \leftarrow u$                            ▷ Pour garder la tracer du chemin le plus court
13:   renvoyer  $d, \Pi$ 

```

---

On se munit d'un type de graphe permettant de modéliser les graphes pondérés statiques :

```

1  let n = 6;;
2  type graph = {size: int; adj: int list array; w: int array array};;

```

---

Les graphes qu'on manipule de taille fixe  $n$ . Les sommets sont représentés par des entiers de 0 à  $n - 1$  et on utilise des listes d'adjacence. Le choix d'un type enregistrement, d'un type `int array` et d'un type `int array array` permet d'accéder directement à un élément du graphe :

- `g.size` est l'ordre du graphe,
- `g.adj.(a)` est la liste des voisins d'un sommet  $a$ ,
- `g.w.(a).(b)` est le poids de l'arête  $(a, b)$ .

C1. Créer un graphe vide d'ordre 6. Les poids seront initialisés à `max_int`, c'est à dire l'entier le plus élevé représentable en machine.

**Solution :**

```

1  let g = {
2      size = n;
3      adj = Array.make n [];
4      w = Array.make_matrix n n max_int
5  };;

```

---

C2. Écrire une fonction de signature `add_edge : graph -> int -> int -> int -> unit` qui permet

**Solution :**

```

1  let add_edge g a b wab =
2      g.adj.(a) <- b::g.adj.(a);

```

---

```

3   g.adj.(b) <- a::g.adj.(b);
4   g.w.(a).(b) <- wab;
5   g.w.(b).(a) <- wab;;

```

C3. Compléter le graphe vide précédemment créé graphe à la fonction `add_edge` pour représenter le graphe de la figure 1.

**Solution :**

```

1 add_edge g 0 1 7;;
2 add_edge g 0 2 1;;
3 add_edge g 1 2 5;;
4 add_edge g 2 4 2;;
5 add_edge g 2 5 7;;
6 add_edge g 1 3 4;;
7 add_edge g 1 4 2;;
8 add_edge g 1 5 1;;
9 add_edge g 3 4 5;;
10 add_edge g 4 5 3;;

```

C4. Appliquer à la main l'algorithme de Dijkstra sur le graphe de la figure 1.

	$\Delta$	a	b	c	d	e	f	$\bar{\Delta}$
	{}	0	7	1	$+\infty$	$+\infty$	$+\infty$	{a, b, c, d, e, f}
	{a}	.	7	1	$+\infty$	$+\infty$	$+\infty$	{b, c, d, e, f}
<b>Solution :</b>	{a, c}	.	6	.	$+\infty$	3	8	{b, d, e, f}
	{a, c, e}	.	5	.	8	.	6	{b, d, f}
	{a, c, e, b}	.	.	.	8	.	6	{d, f}
	{a, c, e, b, f}	.	.	.	8	.	.	{d}
	{a, c, e, b, f, d}	.	.	.	.	.	.	{}

C5. Écrire une fonction de signature `pq_dijkstra : graph -> int -> int -> unit` qui affiche sur la console le chemin le plus court d'un sommet *a* à un sommet *b*. On affichera également la distance entre les deux sommets.

**Solution :**

```

1 let show_path start stop d parents =
2   print_string "Cost -> "; print_int d.(stop); print_newline ();
3   print_string "Path -> ";
4   let rec aux current path =
5     if current = start
6     then List.iter (fun e -> print_int e; print_string " ") path
7     else let father = Hashtbl.find parents current in
8           aux father ( father :: path )
9   in aux stop [ stop ];;

```




FIGURE 1 – Graphe pondéré à valeurs positives pour l'application de l'algorithme de Dijkstra.

```

10
11 let pq_dijkstra g start stop =
12   let pq = make_priority_queue 10 (-1,max_int) in
13   let d = Array.make n max_int in
14   d.(start) <- 0;
15   let parents = Hashtbl.create n in
16   insert pq (start, d.(start));
17   while pq.first_free > 0 do
18     let current = get_min pq in
19     begin
20       if current = stop then show_path start stop d parents;
21       let update x = if d.(x) > d.(current) + g.w.(current).(x)
22       then begin
23         d.(x) <- d.(current) + g.w.(current).(x);
24         insert pq (x, d.(x));
25         Hashtbl.add parents x current;
26       end
27       in List.iter update g.adj.(current)
28     end
29   done;;
30
31 pq_dijkstra g 0 3;;

```