

# Révisions et automatismes

OPTION INFORMATIQUE - TP n° 3.0 - Olivier Reynet

## À la fin de ce chapitre, je sais :

- ☞ implémenter les opérations élémentaires récursives sur les listes
- ☞ utiliser les vecteurs et la programmation impérative
- ☞ implémenter le tri par insertion, le tri fusion et le tri rapide
- ☞ implémenter les opérations élémentaires récursives sur les arbres binaires
- ☞ effectuer une démonstration par récurrence
- ☞ effectuer une démonstration par induction structurelle

## A Listes et récursivité

Écrire des fonctions OCaml qui implémentent les opérations suivantes sur les listes :

- A1. Calculer la longueur d'une liste (`length`).
- A2. Déterminer le *i*-ème élément d'une liste (`at` ou `nth`). On considèrera deux solutions : la première échoue en lançant une exception si l'élément n'est pas trouvé. La seconde renvoie un type `option`.
- A3. Tester l'appartenance d'un élément à une liste (`mem`).
- A4. Concaténer de deux listes (`concat`).
- A5. Aplatir une liste de listes (`flatten`).
- A6. Appliquer une fonction à tous les éléments d'une liste et retourner la liste correspondante : essayer de le faire en temps linéaire (`map`).
- A7. Extraire les deux derniers éléments d'une liste. Cette fonction renvoie une tuple. Fonctionne-t-elle sur une liste de liste?
- A8. Construire la liste inverse d'une liste, c'est à dire que l'élément 0 devient le dernier élément, le premier l'avant dernier...

### Solution :

#### Code 1 – Listes et récursivité

```
1 let test_list = [1;42;99;66;12;7]
2 let test_list_2 = [0;-3;-23]
3 let test_list_list = [[0;1;2]; [3;4]; [5;6;7;8];[9];[10;11]]
4
5 (* not tail recursive *)
6 let rec length l =
```

```
7   match l with
8     | [] -> 0
9     | h::t -> 1 + length t;;
10  length test_list;;
11
12  (* tail recursive *)
13  let length l =
14    let rec aux n = function
15      | [] -> n
16      | _::t -> aux (n + 1) t
17    in
18    aux 0 l;;
19  length test_list;;
20
21  (* raise exception if not found *)
22  let rec at k l =
23    match l with
24      | [] -> failwith "not found !"
25      | h::t when k = 0 -> h
26      | h::t -> at (k - 1) t ;;
27  at 3 test_list;;
28
29  (* return an option *)
30  let rec at k l = match l with
31    | [] -> None
32    | h :: t -> if k = 0 then Some h else at (k - 1) t;;
33  at 3 test_list;;
34
35  let rec mem x l =
36    match l with
37      | [] -> false
38      | h::t when h = x -> true
39      | h::t -> mem x t;;
40  mem 66 test_list;;
41  mem 0 test_list;;
42
43
44  let rec concat l1 l2 =
45    match l1 with
46      | [] -> l2
47      | a::q -> a::(concat q l2) ;;
48  concat test_list test_list_2;;
49
50
51  let rec flatten l =
52    match l with
53      | [] -> []
54      | h::t -> concat h (flatten t) ;;
55  flatten test_list_list;;
56
57  (* @ operator *)
58  let rec flatten l =
59    match l with
60      | [] -> []
61      | h::t -> h @ (flatten t) ;;
```

```

62 flatten test_list_list;;
63
64 let rec iter f l =
65   match l with
66   | [] -> []
67   | h::t -> f(h); iter f t;;
68 iter (fun x -> print_int (x*x); print_newline()) test_list;;
69
70 let rec map f l =
71   match l with
72   | [] -> []
73   | h::t -> f(h)::(map f t) ;;
74 map (fun x -> x*x) test_list;;
75
76 let rec last_two l =
77   match l with
78   | [] | [_] -> failwith "not enough elements !"
79   | [a; b] -> (a,b)
80   | _::t -> last_two t;;
81 last_two test_list;;
82 last_two test_list_list;;
83
84 let rev list =
85   let rec aux built l =
86     match l with
87     | [] -> built
88     | h::t -> aux (h::built) t
89   in
90   aux [] list;;
91 rev test_list;;
92 rev test_list_list;;

```

## B Vecteurs et programmation impérative

- B1. Écrire une fonction qui calcule le n-ième terme de la suite de Fibonacci en utilisant un vecteur où sont stockées les valeurs successives. Quelle est sa complexité temporelle? spatiale?
- B2. Écrire une fonction qui calcule le n-ième terme de la suite de Fibonacci en utilisant des références. Quelle est sa complexité temporelle? spatiale?
- B3. Écrire une fonction `mem x tab` qui test l'appartenance d'un élément `x` à un vecteur `tab`. Quelle est sa complexité? Peut-on faire mieux si l'on suppose le tableau trié? Programmer une version efficace dans ce cas.
- B4. Écrire une fonction `v_concat v1 v2` qui concatène deux tableaux unidimensionnels et renvoie le nouveau tableau ainsi créé.
- B5. Écrire une fonction `m_concat m1 m2` qui concatène deux matrices ayant le même nombre de lignes.
- B6. Écrire une fonction `array_map f tab` qui, sur le modèle de la fonction `map` pour les listes, renvoie un nouveau vecteur contenant les images des éléments de `tab` par la fonction `f`.

**Solution :****Code 2 – Vecteurs**

```
1 (* Suite de Fibonacci - Array *)
2
3 let fibo n =
4   let tab = Array.make (n + 1) 1 in
5   for i = 2 to n do
6     tab.(i) <- tab.(i - 1) + tab.(i - 2)
7   done;
8   tab.(n) ;;
9 fibo 6;;
10
11 (* Suite de Fibonacci - references *)
12
13 let fib n =
14   let u0 = ref 1 and u1 = ref 1 in
15   for i = 2 to n do
16     let tmp = !u0 in (* ne pas écraser une variable !*)
17     u0 := !u1;
18     u1 := tmp + !u1 ;
19   done;
20   !u1 ;;
21 fib 6;;
22
23
24 let test_tab = [|13;66;1;42;17;9;33;11|];;
25 let test_sorted_tab = [|1;9;11;13;17;33;42;66|];;
26
27
28 let mem x tab =
29   let n = Array.length tab and b = ref false and i = ref 0 in
30   while (not !b) && (!i < n) do
31     if tab.(!i) = x then
32       b := true;
33       incr(i) ;
34     done;
35     !b ;;
36 mem 13 test_tab;;
37 mem 2 test_tab;;
38
39
40
41 (* Si tab est trié, recherche dichotomique *)
42
43 let dico_mem x tab =
44   let n = Array.length tab and b = ref false in
45   let g = ref 0 and d = ref (n - 1) in (* indices de début et de fin *)
46   while (not !b) && g <= d do
47     let m = ((!g) + (!d)) / 2 in
48     if tab.(m) = x then
49       b := true
50     else if tab.(m) < x then
51       g := m + 1
52     else
```

```

53     d := m - 1
54     done ;
55     !b ;;
56     dico_mem 13 test_sorted_tab;;
57     dico_mem 2 test_sorted_tab;;
58
59     (* version récursive *)
60     let rec_dico_mem x tab =
61         let rec aux g d =
62             if g > d
63             then false
64             else let m = (g+d)/2 in
65                 if tab.(m) = x
66                 then true
67                 else if tab.(m) < x
68                 then aux (m+1) d
69                 else aux g (m-1)
70         in aux 0 (Array.length tab - 1) ;;
71     rec_dico_mem 13 test_sorted_tab;;
72     rec_dico_mem 2 test_sorted_tab;;
73
74     (* Concaténation de vecteurs *)
75
76     let v_concat v1 v2 =
77         let n1 = Array.length v1 and n2 = Array.length v2 in
78         let a = Array.make (n1 + n2) v1.(0) in
79         for i = 1 to n1 - 1 do
80             a.(i) <- v1.(i)
81         done;
82         for i = 0 to n2 - 1 do
83             a.(n1 + i) <- v2.(i)
84         done;
85         a ;;
86     v_concat test_tab test_sorted_tab;;
87
88     (* Concaténation de matrices *)
89
90     let m_concat m1 m2 =
91         let n1 = Array.length m1 and n2 = Array.length m2 in
92         if n1 <> n2 then
93             failwith "wrong size : can not concat"
94         else
95             let m = Array.make n1 [| |] in
96             for i = 0 to n1-1 do
97                 m.(i) <- v_concat m1.(i) m2.(i)
98             done;
99             m ;;
100     let m1 = [| [|13;66|]; [|1;42;17|]; [|9;33;11|] |];;
101     let m2 = [| [|3;6|]; [|0;41;16|]; [|8;32;10|] |];;
102     let m3 = [| [|0;41;16|]; [|8;32;10|] |];;
103     m_concat m1 m2;;
104     m_concat m1 m3;;
105
106     (* map pour les array *)
107     let array_map f tab =

```

```

108 let n = Array.length tab in
109 let out = Array.make n 0 in
110 for i = 0 to n-1 do
111   out.(i) <- f(tab.(i))
112 done;
113 out ;;
114 array_map (fun x -> x*x) test_tab;;

```

## C Tris

C1. Écrire une fonction qui effectue le tri par insertion. On décompose le problème comme suit :

- Écrire une fonction récursive de prototype `val insert_elem : 'a list -> 'a -> 'a list` qui insère un élément dans une liste triée à la bonne place.
- Écrire une fonction de prototype `val insert_sort : 'a list -> 'a list` qui trie par insertion une liste.
- Utiliser les fonctions `List.fold_left` et `insert_elem` pour écrire une fonction le tri par insertion en une seule ligne.

C2. Discuter la terminaison, la correction et la complexité de la fonction `insert_elem`.

### Solution :

- Terminaison : la fonction `insert_elem` effectue un appel récursif sur une liste dont la taille est strictement plus petite, car on a enlevé un élément. Deux cas sont possibles alors : soit la fonction est appelée sur une liste vide et termine, soit la fonction extrait un élément en tête supérieur à l'élément à insérer et termine. Dans les deux cas, la fonction se termine.
- Correction : on procède par récurrence sur la taille de la liste.

Soit  $\mathcal{P}_n$  la propriété suivante : «pour tout liste `sorted` triée de taille  $n$ , `insert_elem sorted e` renvoie une liste triée dont les éléments sont ceux de `sorted` et  $e$ ».

- Initialisation :  $\mathcal{P}_0$  est vraie car un singleton `[e]` est trié.
- Hérédité : Supposons que  $\mathcal{P}_n$  soit vraie. Soit une liste  $l = h :: t$  de taille  $n+1$  et triée. Si l'élément à insérer  $e$  est plus petit que  $h$ , alors il est inséré en tête et la liste  $e :: h :: t$  est correctement triée, puisque, par hypothèse de récurrence,  $h :: t$  est triée. Si l'élément  $e$  à insérer est plus grand que  $h$ , alors `(insert_elem t e)` est une liste triée puisque  $t$  est de taille  $n$  et que  $\mathcal{P}_n$  est vraie.  $h :: (\text{insert\_elem } t \ e)$  est donc une liste triée. Donc, si  $\mathcal{P}_n$  est vraie,  $\mathcal{P}_{n+1}$  est vraie.
- Conclusion :  $\mathcal{P}_n$  est vrai pour tout  $n$ .

C3. Discuter la terminaison, la correction et la complexité de la fonction `insert_sort`.

### Solution :

- **Terminaison** : la fonction `insert_sort` effectue un appel récursif sur une liste dont la taille est strictement plus petite, car on a enlevé un élément. Ainsi, l'appel aboutit nécessairement sur une liste vide et la fonction termine.
  - **Correction** : on procède par récurrence sur la taille de la liste.  
Soit  $\mathcal{P}_n$  la propriété suivante : «pour tout liste `to_sort` de taille  $n$  et toute liste `sorted` triée, aux `to_sort sorted` renvoie une liste triée dont les éléments sont ceux de `to_sort` et `sorted`».
1. **Initialisation** :  $\mathcal{P}_0$  est vraie car le singleton `[]` est trié et ne contient aucun élément.
  2. **Hérédité** : Supposons que  $\mathcal{P}_n$  soit vraie. Soit une liste  $l = h:t$  de taille  $n+1$ . La fonction effectue l'appel récursif aux `t (insert_elem sorted e)`. Hors, l'insertion de l'élément est correct, donc le second paramètre (`insert_elem sorted e`) est une liste bien triée. Le premier paramètre est de taille  $n$ . Par hypothèse de récurrence, cet appel récursif renvoie donc bien une liste triée. Donc, si  $\mathcal{P}_n$  est vraie,  $\mathcal{P}_{n+1}$  est vraie.
  3. **Conclusion** :  $\mathcal{P}_n$  est vrai pour tout  $n$ .
- **Complexité** : dans le pire des cas, le tableau est trié dans l'ordre inverse et tous les éléments vont parcourir l'intégralité du tableau de gauche lors de l'insertion : la complexité est alors en  $O(n^2)$  puisque `insert_elem` est en  $O(n)$  et qu'on parcourt tous les éléments de la liste dans `insert_sort`. Dans le meilleur des cas, le tableau est trié et la fonction `insert_elem` est à coût constant  $O(1)$  puisque l'élément est déjà inséré à la bonne place. La complexité vaut alors  $O(n)$ .

C4. Écrire une fonction qui réalise le tri fusion. On décompose le problème comme suit :

- (a) une fonction qui sépare une liste en deux,
- (b) une fonction qui réalise la fusion de deux listes triées en une seule liste triée,
- (c) une fonction qui effectue le tri fusion.

C5. Discuter la terminaison, la correction et la complexité du tri fusion.

**Solution :**

1. **Terminaison** : les appels récursifs se font sur des listes dont la taille est strictement plus petite. Ils aboutissent donc nécessairement sur une liste vide, la condition d'arrêt est atteinte et la fonction se termine.
  2. **Correction** : Soit  $\mathcal{P}_n$  la propriété suivante : «pour tout liste  $l$  de taille inférieure ou égale à  $n$ , `merge (merge_sort l1) (merge_sort l2)` renvoie une liste triée dont les éléments sont ceux de  $l$ ».
- (a) **Initialisation** :  $\mathcal{P}_0$  est vraie car le singleton `[]` est trié et ne contient aucun élément.
  - (b) **Hérédité** : Supposons que  $\mathcal{P}_n$  soit vraie. Soit une liste  $l = h:t$  de taille  $n+1$ . La fonction effectue l'appel récursif `merge (merge_sort l1) (merge_sort l2)`. D'après l'hypothèse de récurrence, comme les tailles de `l1` et `l2` sont inférieures à  $n$  puisque créées à partir de `t` qui est de taille  $n$ , les deuxième et troisième paramètres sont des listes bien triées. Si on suppose la fusion des deux listes correctes, alors cet appel récursif renvoie donc bien une liste triée. Donc, si  $\mathcal{P}_n$  est vraie,  $\mathcal{P}_{n+1}$  est vraie.

(c) Conclusion :  $\mathcal{P}_n$  est vrai pour tout  $n$ .

3. Complexité : on a la récurrence suivante  $T(n) = 2T(n/2) + cn$ , car la complexité de la fonction merge est en  $O(n)$ . La complexité du tri fusion est donc de en  $O(n \log(n))$  (dans tous les cas).

C6. Écrire une fonction qui réalise le tri rapide. On décompose le problème comme suit :

- (a) une fonction qui partitionne une liste en deux d'après le pivot,
- (b) une fonction qui effectue le tri rapide.

C7. Discuter la terminaison, la correction et la complexité du tri rapide.

**Solution :** Le pire des cas pour le tri rapide est lorsque le pivot est le plus grand ou le plus petit, et ce à chaque fois qu'on le choisit. Dans ce cas, les deux partitions sont complètement déséquilibrées et la complexité est en  $O(n^2)$ . Ce revient à faire un tri bulle bidirectionnel.

### Code 3 – Tris

```

1 let test_sorted_list = [1;7;12;27;36;42;54;66;73;88;99]
2 let test_list = [1;42;99;66;36;12;7;27;73;36;54;88]
3
4 (* tri par insertion *)
5
6 let rec insert_elem sorted e =
7     match sorted with
8     | [] -> [e]
9     | h::t -> if h < e then h::(insert_elem t e) else e::h::t;;
10 test_list;;
11 insert_elem test_sorted_list 5;;
12
13 let insert_sort l =
14     let rec aux to_sort sorted =
15         match to_sort with
16         | [] -> sorted
17         | e::t -> aux t (insert_elem sorted e)
18     in aux l [];;
19 insert_sort test_list;;
20
21 let fold_left_insert_sort l = List.fold_left insert_elem [] l;;
22 test_list;;
23 fold_left_insert_sort test_list;;
24
25
26
27 (* tri fusion *)
28
29 let rec slice l =
30     match l with
31     | [] -> ([],[])
32     | [a] -> ([a],[])
33     | a::b::t -> let (l1,l2) = slice t in (a::l1, b::l2);;
34
35 let rec merge l1 l2 =

```



```

36  match (l1,l2) with
37  | ([],l2) -> l2
38  | (l1,[]) -> l1
39  | (a1::t1, a2::_) when a1 < a2 -> a1::(merge t1 l2)
40  | (_::_, a2::t2) -> a2::(merge l1 t2) ;;
41
42  let rec merge_sort l =
43    match l with
44    | [] -> []
45    | [a] -> [a]
46    | l -> let (l1,l2) = slice l
47            in merge (merge_sort l1) (merge_sort l2) ;;
48  test_list;;
49  merge_sort test_list;;
50
51
52  (* tri rapide *)
53  let rec partition l pivot=
54    match l with
55    | [] -> [],[]
56    | t::q -> let (l1,l2) = partition q pivot in
57              if (t < pivot) then (t::l1,l2) else (l1,t::l2);;
58
59  let rec quick_sort l =
60    match l with
61    | [] -> []
62    | h::t ->
63      let (l1,l2) = partition t h in
64      (quick_sort l1)@(h::(quick_sort l2));;
65  test_list;;
66  quick_sort test_list;;

```

## D Arbres binaires de recherche

On considère la définition récursive du type 'a arbre :

```
1  type 'a tree = Nil | Node of 'a tree * 'a * 'a tree ;;
```

Écrire les fonctions qui implémentent les opérations suivantes sur un type tree :

- D1. Calculer la hauteur d'un arbre binaire.
- D2. Calculer le nombre total de nœuds d'un arbre binaire.
- D3. Calculer le nombre de feuilles d'un arbre binaire.
- D4. Calculer récursivement le nombre de nœuds internes d'un arbre binaire. Démontrer par induction structurelle que cette fonction est correcte.

**Solution :** Soit  $\mathcal{P}(t)$  : «La fonction `internal_nodes_nb` appliquée à l'arbre  $t$  renvoie le nombre de nœuds internes de l'arbre». La démonstration par induction structurelle procède comme suit :

1. **(CAS DE BASE)** On considère les cas de base. Tout d'abord, l'arbre vide `Nil`. La fonction renvoie 0 qui est le nombre de nœuds (internes) de l'arbre. L'autre cas de base est l'arbre ne possédant qu'un seul nœud (`Nil`, `_`, `Nil`). Dans ce cas, la fonction renvoie 0, ce qui est exact, puisque ce nœud est aussi une feuille.
2. **(PAS D'INDUCTION)** Soit  $e \in E$  une étiquette et  $f_g \in \mathcal{A}$  et  $f_d \in \mathcal{A}$  deux arbres binaires pour lesquels  $\mathcal{P}(f_g)$  et  $\mathcal{P}(f_d)$  sont vraies : la fonction `internal_nodes_nb` renvoie le nombre de nœuds internes de ces arbres. Pour l'arbre construit selon la règle de la structure, c'est à dire  $(f_g, e, f_d)$ , la fonction renvoie un de plus que le nombre de nœuds des fils. Ce qui est exact puisqu'on a ajouté un nœud à l'arbre. Donc,  $\mathcal{P}(f_g, e, f_d)$  est vraie.
3. **(CONCLUSION)** Quelque soit l'arbre  $t$ ,  $\mathcal{P}(t)$  est vraie.

D5. Ajouter un élément à la fin de la branche la plus à gauche et renvoyer le nouvel arbre.

### Solution :

#### Code 4 – Arbres binaires

```

1 type 'a tree = Nil | Node of 'a tree * 'a * 'a tree ;;
2
3 let t = Node(
4     Node(Node(Nil, 3, Nil), 7, Node( Nil, 9, Nil)),
5     13,
6     Node(Node(Nil, 4, Node(Nil, 2, Nil)), 17, Node(Nil, 7, Nil))
7 );;
8
9 let rec height a =
10     match a with
11     | Nil -> -1
12     | Node(fg,_,fd) -> 1 + max (height fg) (height fd) ;;
13 height t;;
14
15 let rec size a =
16     match a with
17     | Nil -> 0
18     | Node(fg, x, fd) -> 1 + size fg + size fd ;;
19 size t;;
20
21 let rec leaves_nb a =
22     match a with
23     | Nil -> 0
24     | Node(Nil, x, Nil) -> 1
25     | Node(fg, x, fd) -> leaves_nb fg + leaves_nb fd ;;
26 leaves_nb t;;
27
28 let internal_nodes_nb a = size a - leaves_nb a ;;
29 internal_nodes_nb t;;
30
31 let rec internal_nodes_nb a =
32     match a with
33     | Nil -> 0
34     | Node(Nil,_,Nil) -> 0
35     | Node(fg,x,fd) -> 1 + internal_nodes_nb fg + internal_nodes_nb fd ;;
```

```
36 internal_nodes_nb t;;
37
38
39 let rec add e a =
40   match a with
41   | Nil -> Node(Nil,e,Nil)
42   | Node(fg,x,fd) -> Node((add e fg), x, fd) ;;
43 add 23 t;;
```

---